# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Maintainer: Charles Mauch <cmauch@gmail.com>

pkgname=perl-net-jabber
pkgver=2.0
pkgrel=13
pkgdesc="Perl/CPAN Module Net::Jabber : Access to the Jabber protocol"
arch=("any")
url="https://search.cpan.org/dist/Net-Jabber"
license=("GPL" "PerlArtistic")
makedepends=("perl-net-xmpp")
depends=("perl-net-xmpp")
source=("https://www.cpan.org/authors/id/R/RE/REATMON/Net-Jabber-$pkgver.tar.gz")
sha256sums=('dfba394a6166a5a5e35ce763f408f97788047cc39ce5acad802e281432c88011')

build() {
  cd "$srcdir"/Net-Jabber-$pkgver
  PERL_MM_USE_DEFAULT=1 perl Makefile.PL INSTALLDIRS=vendor
  make
}

package() {
  cd "$srcdir"/Net-Jabber-$pkgver
  make install DESTDIR="$pkgdir"
  find "$pkgdir" -name '.packlist' -delete
  find "$pkgdir" -name '*.pod' -delete
}
